module ad_dds_cordic_pipe#(

  // parameters

  // Range = N/A
  parameter P_DW = 16,
  // Range = N/A
  parameter D_DW = 16,
  // Range = N/A
  parameter DELAY_DW = 1,
  // Range = 0-(DW - 1)
  parameter SHIFT = 0) (

  // Interface

                      input                          clk,
  (* keep = "TRUE" *) input                          dir,
  (* keep = "TRUE" *) input             [  D_DW-1:0] dataa_x,
  (* keep = "TRUE" *) input             [  D_DW-1:0] dataa_y,
  (* keep = "TRUE" *) input             [  P_DW-1:0] dataa_z,
  (* keep = "TRUE" *) input             [  P_DW-1:0] datab_z,
  (* keep = "TRUE" *) output reg        [  D_DW-1:0] result_x,
  (* keep = "TRUE" *) output reg        [  D_DW-1:0] result_y,
  (* keep = "TRUE" *) output reg        [  P_DW-1:0] result_z,
                      input             [DELAY_DW:1] data_delay_in,
                      output            [DELAY_DW:1] data_delay_out);

  // Registers Declarations

  reg   [DELAY_DW:1] data_delay = 'd0;

  // Wires Declarations

  wire  [  D_DW-1:0]  sgn_shift_x;
  wire  [  D_DW-1:0]  sgn_shift_y;
  wire                dir_inv = ~dir;

  // Sign shift

  assign sgn_shift_x = {{SHIFT{dataa_x[D_DW-1]}}, dataa_x[D_DW-1:SHIFT]};
  assign sgn_shift_y = {{SHIFT{dataa_y[D_DW-1]}}, dataa_y[D_DW-1:SHIFT]};

  // Stage rotation

  always @(posedge clk) begin
    result_x <= dataa_x + ({D_DW{dir_inv}} ^ sgn_shift_y) + dir_inv;
    result_y <= dataa_y + ({D_DW{dir}}     ^ sgn_shift_x) + dir;
    result_z <= dataa_z + ({P_DW{dir_inv}} ^     datab_z) + dir_inv;
  end

  // Delay data (if used)

  generate
    if (DELAY_DW > 1) begin
      always @(posedge clk) begin
        data_delay <= data_delay_in;
      end
    end
  endgenerate

  assign data_delay_out = data_delay;

endmodule